The Hopf fibration is too mainstream. Instead of covering the 3-sphere with smooth curves, let's cover it with surfaces.

The Reeb foliation of the 3-sphere has a relatively simple explanation:
- Embedding a central torus C in the 3-sphere partitions it into two solid tori
- Within each solid torus, the leaves are all identical, just rotated copies of each other. Each can be interpreted as the surface of revolution of a curve that starts with a vertical tangent then tends to a horizontal asymptote, but wrapped around itself inside the torus (see below):

```
c = 0.4; (* determines the ratio of the torus' tube radius to full radius *)
F[x_] := Sqrt[2/π ArcTan[x]]; (* The leaf's shape function *)
TorusPoints[θ_, α_, f_] :=
  {Cos[θ], Sin[θ], 0} + c f[θ] (Cos[α] {Cos[θ], Sin[θ], 0} + Sin[α] {0, 0, 1});
 (* Plotting points around the torus but with varying
 depth parameter f depending on θ *)

Plot[{F[x], 1}, {x, 0, 5}, PlotRange -> All]
ParametricPlot3D[{x, Sin[t] F[x], Cos[t] F[x]},
 {x, 0, 5}, {t, 0, 2 π}, Mesh → None, MaxRecursion → 5]
ParametricPlot[{TorusPoints[θ, 0, F][[1 ;; 2]], {Cos[θ], Sin[θ]},
  (1 + c) {Cos[θ], Sin[θ]}}, {θ, 0, 2.5 π}, Mesh → None]
ParametricPlot3D[TorusPoints[θ, α, F], {θ, 0, 2.5 π}, {α, 0, 2 π},
 PlotStyle → Opacity[0.5], Mesh → None, MaxRecursion → 5]
```




0
2
4
0.5
0.0
−0.5
0.5
0.0
−0.5


1.5
1.0
0.5
−0.5
−1.0
−1.5
−1.5
−1.0
−0.5
0.5
1.0
1.5


0.4
0.2
0.0
−0.2
−0.4
−1
0
1
1
0
−1

To visualize the Reeb foliation, we will use the stereographic projection of the 3-sphere onto $\mathbb{R}^3$. We will set up a standard torus C and foliate the inside as above.

```
insides = Table[ParametricPlot3D[TorusPoints[θ, α, F[# + k] &], {θ, -k, π - k},
    {α, -π, 0}, Mesh → None, PlotStyle → {Hue[k / 30], Specularity[White, 100]},
    MaxRecursion → 5], {k, 0, 2 π - π / 6, π / 6}];
sampleinside = ParametricPlot3D[TorusPoints[θ, α, F], {θ, 0, 4 π / 3}, {α, -π, π},
   Mesh → None, PlotStyle → {Hue[0], Specularity[White, 100]}, MaxRecursion → 5];
Ctorus = ParametricPlot3D[TorusPoints[θ, α, 1 &], {θ, 0, 2 π}, {α, -π - .4, .4},
   Mesh → None, PlotStyle → {Blue, Specularity[White, 100]}, MaxRecursion → 5];
```

```
Show[{sampleinside, insides, Ctorus}, PlotRange → All, Boxed → False, Axes → False]
```



Foliating the outer solid torus takes a bit more thinking. We had a nice parametrization of the inside of C; now we need one for the exterior. One parameter will simply be angle around the z-axis, but then we have to parametrize a half-plane-minus-a-circle (the cross-section of the exterior of C at a given angle around the z-axis). The idea is that we can use a Mobius transformation to parametrize this space by comparing it to a suitable annulus.

The function Mobius takes the unit circle to the vertical axis:

```
Mobius[{x_, y_}] := {Re[((x + I y) - 1)/((x + I y) + 1)], Im[((x + I y) - 1)/((x + I y) + 1)]};
```

We can identify the annulus we need by finding the preimage of three points on the circle to be removed from the half-plane; it is straightforward to see that the circle centered at 1 with radius c comes from the circle centered at -1 with radius 2/c. Thus, we parametrize the annulus between this circle and the unit circle (the level sets must meet the unit circle at right angles), and push the parametrization forward using the Mobius function.

```
AnnulusParam[β_, t_] := (maxt = -Cos[β] + Sqrt[(2/c)^2 - Sin[β]^2];
  (1 + t (maxt - 1)) {Cos[β], Sin[β]})
```

Illustrating the parametrization of the annulus, and how it transforms into a parametrization of the half-plane-minus-disk. Also included are the desired cross-sections of leaves. Note that they must meet the z-axis at right angles in order for their surface of revolution to be smooth.

```
coordEx =
  Table[{AnnulusParam[#1, #2], 0, 2 π, {LightGray}} &[xx, t], {t, 0, 1, 1/6}] ~Join~
   Table[{AnnulusParam[#1, #2], 0, 1, {LightGray}} &[β, xx], {β, 0, 2 π, π/8}] ~Join~
   Table[{AnnulusParam[#1, F[#1 + #2]], If[#2 == π, -1.01 π, -#2],
       2 π - #2, {Hue[#2/(2 π)]}} &[xx, k], {k, 0, 2 π - π / 4, π / 4}];
Show[
 ParametricPlot[#[[1]] /. xx → x, {x, #[[2]], #[[3]]}, PlotStyle → #[[4]]] &/@ coordEx,
 PlotRange → All]
Show[ParametricPlot[Mobius[#[[1]]] /. xx → x, {x, #[[2]], #[[3]]},
    PlotStyle → #[[4]]] &/@ coordEx, PlotRange → {{0, 1 + 2 c}, {-3 c, 3 c}}]
```





We can finally define a function OuterTorusPoints that will allow us to plot the outer leaves, and then we simply need to put all the pieces together!

```
OuterTorusPoints[θ_, β_, f_] := {Cos[θ] Mobius[AnnulusParam[β, f[β]]] [[1]],
  Sin[θ] Mobius[AnnulusParam[β, f[β]]] [[1]], Mobius[AnnulusParam[β, f[β]]] [[2]]};

qual = 8; (* 2 for quick results. Up to 8 for very high resolution *)

leaves = {-0.93 π, -0.8 π, -0.65 π, -0.45 π, 0, 0.55 π, 0.7 π, 0.84 π, 0.93 π, π};
outers = ParametricPlot3D[
    OuterTorusPoints[θ, β, Function[x, F[x + #]]], {θ, 0, π}, {β, -#, 2 π},
    Mesh → None, PlotStyle → {Hue[#/ (10 π) + 1 / 3], Specularity[White, 100]},
    MaxRecursion → qual] & /@ leaves;

insides2 = Table[ParametricPlot3D[TorusPoints[θ, α, F[# + k] &], {θ, -k, π - k},
   {α, -π - .4, -.4}, Mesh → None, PlotStyle → {Hue[k / 30], Specularity[White, 100]},
   MaxRecursion → qual], {k, 0, 2 π - π / 6, π / 6}];

sampleinside2 = ParametricPlot3D[
  TorusPoints[θ, α, F[# + π / 6] &], {θ, -π / 6, 4 π / 3}, {α, -π, π}, Mesh → None,
  PlotStyle → {Hue[Pi / 90], Specularity[White, 100]}, MaxRecursion → qual];
Ctorus2 = ParametricPlot3D[TorusPoints[θ, α, 1 &], {θ, 0, 2 π}, {α, -π + .2, -.2},
  Mesh → None, PlotStyle → {Blue, Specularity[White, 100]}, MaxRecursion → qual];
```

I could use a new desktop background... (“Rasterize” added by DBN to decrease file size)

```
Show[{insides2, outers, sampleinside2, Ctorus2},
  PlotRange → {{-8, 4}, {-4, 8}, {-4, 4}}, Boxed → False,
  Axes → False, Background → Gray, ImageSize → {1296, 768},
  ViewVector → {{10, -20, 10}, {-0.9, 0, 0}}, ViewAngle → π/24] // Rasterize
```